

# 1989年广西青少年程序设计 竞赛试题及参考解答

罗海鹏 苏芳来

(广西计算中心) (广西区电化教育馆)

## 摘 要

广西青少年计算机知识竞赛于1989年5月在南宁举行,本文列出了八道试题及其参考解答,这些试题有一定难度并限定二小时内苹果II机上完成。

一、找出所有不大于50的正整数N,它能被不大于N的平方根的所有整数整除。

解:

让N从1循环到50,用小于等于N的平方根的所有整数去除N,如果都能够除尽,则N是一个所要求的数。

变量意义:

N: 被考查的数;

S: N的平方根;

I: 除数;

T: 临时工作单元。

程序如下:

```
10 FOR N=1 TO 50
20 S=INT(SQR(N))
30 FOR I=2 TO S
40 T=INT(N/I)
50 IF ABS(N-T*I)>0.001 THEN 80
60 NEXT I
70 PRINT N, " ",
80 NEXT N
90 END
RUN
```

1 2 3 4 6 8 12 24

二、Fibonacci数列如下:

0, 1, 1, 2, 3, 5, 8, 13, 21, ……

那从第三个数起, 每一个数均为它前面两个数之和。问在这个数列中, 第一个末四位皆为零的数(不是第一个数), 是数列的第几个数?

解:

我们用变量向后挪动的方法, 生成这个数列的各项。当某一项大于10000时, 则减去10000, 再用差生成新的项。当某一项正好等于10000时, 则找到了所要求的项。

变量意义:

A、B、C: Fibonacci数列中的连续三项,

N: 数列元素的序号。

程序如下:

```

10 A = 0 : B = 1 : N = 2
20 N = N + 1
30 C = A + B
40 IF C = 10000 THEN 80
50 IF C > 10000 THEN C = C - 10000
60 A = B : B = C
70 GOTO 20
80 PRINT N
90 END
RUN
7501

```

三、2的多少次方的末M位数字仅由1和2组成? 输出M = 5时的情况。

解:

用M个元素的数组装2的幂的末M位数字。乘以2时, ☆每一个数组元素都乘以2, 若超过了10, 则减10进1。判断这M个数字是否都是由1或2组成的, 若不是, 则返回☆。

变量意义:

M: 题目中的M,

A: 数组, 每一个装2的幂的一位数字。

N: 2的次方数。

程序如下:

```

10 INPUT M
20 DIM A(M+1)
30 N = 1 : A(1) = 2
40 N = N + 1
50 FOR I = 1 TO M
60 A(I) = A(I) * 2
70 NEXT I
80 FOR I = 1 TO M

```

```

90 IF A(I) >= 10 THEN A(I) = A(I) - 10 : A(I+1) = A(I+1) + 1
100 NEXT I
110 A(M+1) = 0
120 FOR I=1 TO M
130 IF A(I) <> 1 AND A(I) <> 2 THEN 40
140 NEXT I
150 PRINT "2^", N, "=.....",
160 FOR I=M TO 1 STEP -1
170 PRINT A(I),
180 NEXT I
190 END
RUN

```

? 5 ↘

2<sup>589</sup> = .....22112

四、求 1 2 3 4 5 6 7 8 9 1 2 3 4 5 6 7 8 9.....1 2 3 4 5 6 7 8 9 (共11组) 除以 7 8 9 的商和余数。

解:

模拟竖式除法。每除得一位数字的商, 则输出, 再把余数乘以10, 加上被除数的下一个数字, 继续除下去。

变量意义:

T: 商,

R: 余数,

V: 被除数的每一个数字。

程序如下:

```

10 PRINT "123.....789/789=" ;
20 T=INT(1234/789)
30 PRINT T,
40 R=1234-T*789
50 V=4
60 FOR I=1 TO 11
70 V=V+1
80 IF V>9 THEN V=0:GOTO 140
90 R=R*10+V
100 T=INT(R/789)
110 PRINT T,
120 R=R-T*789
130 GOTO 70
140 NEXT I

```

```

150 PRINT
160 PRINT "R=" ; R
170 END

```

RUN

```

1 2 3.....7 8 9 / 7 8 9 = 1 5 6 4 7 2 4 8 3 0 4 6 2 0 6 3 2 3 3 5 0 3 8 8 8 3 2
8 5 0 0 9 2 0 5 4 9 0 9 3 2 4 1 9 4 0 7 2 8 2 2 2 9 2 9 3 2 1 3 7 8 8 2 2 8
5 7 5 0 0 4 9 5 8 6 1 3 0 2 7 7 7 2 3 0 1 2 4 0 7 0 0 0 5 7 8
R = 747

```

五、验证100以内的所有正整数都可以表示为不多于五个正整数的平方和。

解：

求出1到10的平方，放入数组。从1到100循环，看每一个数最大可以容纳哪一个数的平方，剩下的部分又最大可以容纳哪一个数的平方，……例如： $91 = 81 + 9 + 1$ 。

变量意义：

A：数组，装1, 2, ……，10的平方，

B：原是被验证的数，逐渐变为零。

程序如下：

```

10 FOR I=1 TO 10
20 A(I) = I*I
30 NEXT I
40 FOR I=1 TO 100
50 B=I
60 PRINT I, "=",
70 FOR J=1 TO 10
80 IF B<A(J) THEN N=J-1:GOTO 110
90 NEXT J
100 N=10
110 PRINT A(N),
120 B=B-A(N)
130 IF B<>0 THEN PRINT "+", :GOTO 70
140 PRINT
150 NEXT I
160 END

```

RUN

运行结果略去。

六、键盘输入五个不同的数字（不包括零），不许重复，它们能组成多少个一位至五位的是3的倍数的数？

解：

键盘输入五个数字，考查由它们组成的所有的1)一位数、四位数，2)二位数，3)三

位数，4)五位数。一位数一找出就是1个，二位数一找出就是2个，三位数是6个，四位数是24个，五位数是120个。

变量意义：

A：数组，键盘输入五个数字；

Z：五个数字之和；

S：能被3除尽的数的个数；

T：部分数字之和。

程序如下：

```
10 Z=0
20 FOR I=1 TO 5
30 INPUT A(I)
40 Z=Z+A(I)
50 NEXT I
60 S=0
70 FOR I=1 TO 5
80 Y=Z-A(I)
90 IF INT(A(I)/3)*3=A(I) THEN S=S+1
100 IF INT(Y/3)*3=Y THEN S=S+24
110 NEXT I
120 FOR I=1 TO 4
130 FOR J=I+1 TO 5
140 T=A(I)+A(J)
150 IF INT(T/3)*3=T THEN S=S+2
160 NEXT J
170 NEXT I
180 FOR I=1 TO 3
190 FOR J=I+1 TO 4
200 FOR K=J+1 TO 5
210 T=A(I)+A(J)+A(K)
220 IF INT(T/3)*3=T THEN S=S+6
230 NEXT K
240 NEXT J
250 NEXT I
260 IF INT(Z/3)*3=Z THEN S=S+120
270 PRINT "S=", S
280 END
RUN
? 1 ↵
```

? 2 ↘

? 3 ↘

? 4 ↘

? 5 ↘

S = 177

七、用从1到9这九个数字组成3个三位数，并使此三数之比值为1:2:3。找出所有可能的组合方式。

解：

这三个数的第一个数，它的第一位数字只能是1、2或3。取遍所有的组合方式，把所有可能的第一位数字是1、2或3的三位数乘以2、乘以3，如果新组成的两个三位数的数字都和原来的三位数的数字不重复，并且互相也不重复，则这三个数是一组解答。

变量意义：

B，布尔量数组， $B(5) = 1$ 表示数字5已被使用，

M1、M2、M3，3个三位数；

T，分离三位数成三个数字。

程序如下：

```

5 DIM B(12)
10 FOR I=1 TO 3
20 FOR J=1 TO 9
30 B(I) = 1
40 IF B(J) = 1 THEN 350
50 FOR K=1 TO 9
60 B(J) = 1
70 IF B(K) = 1 THEN 340
80 B(K) = 1
90 M1 = I * 100 + J * 10 + K
100 M2 = M1 * 2
110 T = M2
120 T(1) = INT(T/100)
130 T = T - T(1) * 100
140 T(2) = INT(T/10)
150 T(3) = T - T(2) * 10
160 FOR L=1 TO 3
170 IF B(T(L)) = 1 THEN 310
180 B(T(L)) = 1
190 NEXT L
200 M3 = M1 * 3
210 T = M3

```

```

220 T(1) = INT(T/100)
230 T = T - T(1) * 100
240 T(2) = INT(T/10)
250 T(3) = T - T(2) * 10
260 FOR L=1 TO 3
270 IF B(T)(L) = 1 THEN 310
280 B(T(L)) = 1
290 NEXT L
300 PRINT M1, ",", M2, ",", M3
310 FOR L=1 TO 9
320 B(L) = 0
330 NEXT L
340 NEXT K
350 NEXT J
360 NEXT I
370 END
RUN
192, 384, 576
219, 438, 657
273, 546, 819
327, 654, 981

```

八、验证100以内的所有正整数都可以表示为不多于4个正整数的平方和。

解:

把1, 2, …, 10的平方装入数组。在从1到100的大循环里,看循环变量是否是数组中的某一个数;如果不是,看是否是数组中的某两个数之和;如果不是,再看是否是数组中的某三个数之和;如果还不是,再看是否是数组中的某四个数之和。

变量意义:

A: 10个元素的数组,装1至10的平方。

程序如下:

```

10 FOR I=1 TO 10
20 A(I) = I*I
30 NEXT I
40 FOR I=1 TO 100
50 FOR J1=1 TO 10
60 IF I=A(J1) THEN PRINT I, "=", A(J1):GOTO 510
70 IF I<A(J1) THEN 90
80 NEXT J1
90 FOR J1=10 TO 1 STEP -1

```

```
100 IF A(J1) > I THEN 180
110 T = A(J1)
120 FOR J2 = 10 TO 1 STEP -1
130 T = T + A(J2)
140 IF T > I THEN T = T - A(J2) : GOTO 170
150 IF T < I THEN 170
160 PRINT I, "=", A(J1), "+", A(J2) : GOTO 510
170 NEXT J2
180 NEXT J1
190 FOR J1 = 10 TO 1 STEP -1
200 IF A(J1) > I THEN 320
210 T = A(J1)
220 FOR J2 = 10 TO 1 STEP -1
230 T = T + A(J2)
240 IF T > I THEN T = T - A(J2) : GOTO 310
250 FOR J3 = 10 TO 1 STEP -1
260 T = T + A(J3)
270 IF T < I THEN T = A(J1) : GOTO 310
280 IF T = I THEN PRINT I, "=", A(J1), "+", A(J2),
    "+", A(J3) : GOTO 510
290 T = T - A(J3)
300 NEXT J3
310 NEXT J2
320 NEXT J1
330 FOR J1 = 10 TO 1 STEP -1
340 IF A(J1) > I THEN 500
350 T = A(J1)
360 FOR J2 = 10 TO 1 STEP -1
370 T = T + A(J2)
380 IF T > I THEN T = T - A(J2) : GOTO 490
390 FOR J3 = 10 TO 1 STEP -1
400 T = T + A(J3)
410 IF T > I THEN T = T - A(J3) : GOTO 480
420 FOR J4 = 10 TO 1 STEP -1
430 T = T + A(J4)
440 IF T < I THEN T = A(J1) + A(J2) : GOTO 480
450 IF T = I THEN PRINT I, "=", A(J1), "+", A(J2),
    "+", A(J3), "+", A(J4) : GOTO 510
```

460 T = T - A ( J4 )  
470 NEXT J4  
480 NEXT J3  
490 NEXT J2  
500 NEXT J1  
510 NEXT I  
520 END

## THE QUESTIONS AND REFERENCE SOLUTIONS OF GUANGXI YOUNGSTERS PROGRAMMING IN 1989

Luo Haipeng

(*Computer Centre of Guangxi*)

Su Fanglai

(*Audie-Visual Educational Centre of Guangxi*)

### ABSTRACT

Guangxi youngsters competition of computer knowledge including programming was held in Nanning in May, 1989. In this article, we present eight questions of the programming competition which mostly involve numbers of specific property and reference solutions. These questions are moderately difficult and prescribe two hours for fulfilment on the APPLEII computer.